

Supplementary Information S4 - Implementation of the joint model on Monolix software

Creating data set

The data set is composed of three parts: the longitudinal one and the two survival ones. We denote *Obs_id* the observation id, 1 for longitudinal data, 2 for time-to-death data, and 3 for time-to-discharge data. We suppose that patient A was discharged from ICU at day 7. For this patient, the data has the following form:

Patient	Time	Obs	Obs_id	COV
A	0	11	1	2
A	1	11	1	2
A	2	12	1	2
A	3	9	1	2
A	4	8	1	2
A	5	7	1	2
A	6	8	1	2
A	7	7	1	2
A	0	0	2	2
A	30	0	2	2
A	0	0	3	2
A	7	1	3	2

Because we are in a subdistribution approach, the assumption is that the patient will never experience the other event (death in ICU), that is the reason why the event with observation id 2 was not observed until the end of the study (at time 30).

Model file

If we consider the model presented in Method section, the model file is written as:

```
DESCRIPTION : Joint Model

[LONGITUDINAL]
input={beta0,b1,b2,a,tlag,h1,h2,alpha1,alpha2,h_cov1,h_cov2}

EQUATION :

SOFA = max(min(beta0 + a*(exp(b1*(t-tlag))-exp(b2*(t-tlag))),24),0)

haz1 = h1*exp(alpha1*SOFA + h_cov1)
haz2 = h2*exp(alpha2*SOFA + h_cov2)

DEFINITION:
death = {type=event, maxEventNumber= 1, hazard=haz1}
discharge = {type=event, maxEventNumber=1, hazard=haz2}

OUTPUT:
output={SOFA, death, discharge}
```

Mlxtran file

```
<DATAFILE>

[FILEINFO]
file = './data.txt'
delimiter = space
header = {id, Time, Obs, Obs_id, age}

[CONTENT]
id = {use=identifier}
Time = {use=time}
Obs = {use=observation, name={y1, y2, y3}, yname={'1', '2', '3'}, type={continuous, event, event}}
Obs_id = {use=observationtype}
age = {use=covariate, type=categorical}

<MODEL>

[COVARIATE]
input = age

age = {type=categorical, categories={1, 2, 3}}

[INDIVIDUAL]
input = {beta0_pop, omega_beta0, b1_pop, omega_b1, b2_pop, omega_b2, a_pop, omega_a, tlag_pop, omega_tlag, h1_pop, h2_pop,
        alpha1_pop, alpha2_pop, h_cov1_pop, h_cov2_pop, age, beta_h_cov1_age_2, beta_h_cov1_age_3, beta_h_cov2_age_2, beta_h_cov2_age_3}

age = {type=categorical, categories={1, 2, 3}}

DEFINITION:
beta0 = {distribution=normal, typical=beta0_pop, sd=omega_beta0}
b1 = {distribution=normal, typical=b1_pop, sd=omega_b1}
b2 = {distribution=normal, typical=b2_pop, sd=omega_b2}
a = {distribution=logNormal, typical=a_pop, sd=omega_a}
tlag = {distribution=normal, typical=tlag_pop, sd=omega_tlag}
h1 = {distribution=logNormal, typical=h1_pop, no-variability}
h2 = {distribution=logNormal, typical=h2_pop, no-variability}
alpha1 = {distribution=normal, typical=alpha1_pop, no-variability}
alpha2 = {distribution=normal, typical=alpha2_pop, no-variability}
h_cov1 = {distribution=normal, typical=h_cov1_pop, covariate=age, coefficient={0, beta_h_cov1_age_2, beta_h_cov1_age_3}, no-variability}
h_cov2 = {distribution=normal, typical=h_cov2_pop, covariate=age, coefficient={0, beta_h_cov2_age_2, beta_h_cov2_age_3}, no-variability}

[LONGITUDINAL]
input = {sigma_a, sigma_b}

file = './model.txt'

DEFINITION:
y1 = {distribution=normal, prediction=SOFA, errorModel=combined1(sigma_a, sigma_b)}

<FIT>
data = {y2, y3, y1}
model = {deces, sortie, y1}

<PARAMETER>|
sigma_a = {value=1, method=MLE}
a_pop = {value=1, method=MLE}
alpha1_pop = {value=0, method=MLE}
alpha2_pop = {value=0, method=MLE}
sigma_b = {value=1, method=MLE}
b1_pop = {value=1, method=MLE}
b2_pop = {value=1, method=MLE}
beta0_pop = {value=1, method=MLE}
beta_h_cov1_age_2 = {value=0, method=MLE}
beta_h_cov1_age_3 = {value=0, method=MLE}
beta_h_cov2_age_2 = {value=0, method=MLE}
beta_h_cov2_age_3 = {value=0, method=MLE}
h1_pop = {value=1, method=MLE}
h2_pop = {value=1, method=MLE}
h_cov1_pop = {value=0, method=FIXED}
h_cov2_pop = {value=0, method=FIXED}
omega_a = {value=1, method=MLE}
omega_b1 = {value=1, method=MLE}
omega_b2 = {value=1, method=MLE}
omega_beta0 = {value=1, method=MLE}
omega_tlag = {value=1, method=MLE}
tlag_pop = {value=1, method=MLE}

<MONOLIX>
```

```
[TASKS]
populationParameters()
individualParameters(method = {conditionalMean, conditionalMode })
fim(method = StochasticApproximation)
loglikelihood(method = ImportanceSampling)
plotResult(method = {outputplot, indfits, obspred, parameterdistribution, covariateModeldiagnosis, randomeffects,
                    |covarianceModeldiagnosis, saemresults })

[SETTINGS]
GLOBAL:
exportpath = './'

POPULATION:
smoothingiterations = 2000
exploratoryiterations = 2000

INDIVIDUAL:
optimizationiterations = 2000

FIM:
maxiterations = 2000
```